Module 5 Oplossen van stelsels vergelijkingen

Onderwerp	Stelsels vergelijkingen.
Voorkennis	—
Expressies	`lhs, rhs, assign, isolate, solve, identity, RootOf, allvalues, fsolve, avoid`
Zie ook	Module 3, 8, 14 en 25.

5.1 Stelsels vergelijkingen

In deze module houden we ons bezig met het oplossen van stelsels vergelijkingen van de vorm

$$\left\{\begin{array}{rcl} f_1(x_1,\ldots,x_n) & = & 0 \\ \vdots & & \\ f_k(x_1,\ldots,x_n) & = & 0 \end{array}\right. \tag{5.1}$$

Hierbij zijn $f_1,\ldots,f_k$ functies van n variabelen; (5.1) is een stelsel van k vergelijkingen met n onbekenden.
'Een oplossing' (enkelvoud!) van het stelsel vergelijkingen is een verzameling bij elkaar horende waarden voor $x_1,\ldots,x_n$ die ingevuld in $f_1,\ldots,f_k$ steeds 0 opleveren.
Het oplossen van een stelsel als in (5.1) lukt bijna nooit als de functies $f_1,\ldots,f_k$ niet-lineair zijn.

5.2 Vergelijkingen in Maple

Elke expressie waarin één gelijkteken voorkomt wordt door Maple opgevat als een vergelijking. We kunnen een vergelijking ook een naam geven met de toekenningsoperator `:=`. Zo geven we met

```
v := x = 5;
```

de *vergelijking* '$x = 5$' de *naam* v. Door deze toekenning heeft de variabele `x` géén waarde gekregen, en de variabele `v` wél (namelijk:
`rhs` $x = 5$). Het *rechterlid* daarvan wordt verkregen met `rhs(v)` (right
`lhs` hand side). Voor het linkerlid kan men uiteraard het commando `lhs` gebruiken.

assign

isolate

Ook is het mogelijk (maar vrijwel nooit nuttig) om de waarde 5 toe te kennen aan de variabele x door middel van `assign(v)`. Dus `assign(x=5)` heeft precies dezelfde betekenis als `x:=5`.

Het commando `isolate` biedt nog een mogelijkheid om een vergelijking enigszins te manipuleren. Bijvoorbeeld, als `v1` de vergelijking $x^3 + 3x^2 - x - 1 = 0$ is, dan resulteert

```
v2 := isolate( v1, x^2 );
```

in

$$v2 := x^2 = -\frac{1}{3}x^3 + \frac{1}{3}x + \frac{1}{3}\,.$$

De mogelijkheden zijn echter beperkt. Zo zal

```
v3 := isolate( v1, x+1 );
```

de foutboodschap geven dat de uitdrukking `x+1` niet in de vergelijking voorkomt.

Een *stelsel* vergelijkingen is in Maple een *verzameling* van vergelijkingen. We geven dat aan door de vergelijkingen, door komma's gescheiden, tussen accolades te plaatsen.

5.3 Eén vergelijking met één onbekende

solve

Voor het oplossen van een stelsel vergelijkingen heeft Maple de procedure `solve`. We beginnen met een voorbeeld van één vergelijking met één onbekende:

Voorbeeldopgave

Vind de nulpunten van $p(x) = x^3 - 5x^2 - 3x + 15$;

Idem voor $f(x) = e^x$.

Voorbeeldsessie

```
> p := x^3-5*x^2-3*x+15;
```

$$p := x^3 - 5\,x^2 - 3\,x + 15$$

De nulpunten van p worden gevonden met:

```
> s := solve(p);
```

$$s := 5,\ \sqrt{3},\ -\sqrt{3}$$

Er zijn drie nulpunten met de namen **s[1]**, **s[2]** en **s[3]**.

```
> s[2];
```

$$\sqrt{3}$$

Als we de vergelijking en de op te lossen variabelen tussen accolades zetten, dan komen de oplossingen in een vorm die geschikt is voor het **subs**-commando

```
> s := solve( {p=0}, {x} );
```

$$s := \{x = 5\}\,, \left\{x = \sqrt{3}\right\}, \left\{x = -\sqrt{3}\right\}$$

We controleren de tweede oplossing door hem te substitueren in p:

```
> subs( s[2], p );
```

$$0$$

We kennen deze waarde toe aan de variabele X (hoofdletter!)

```
> X := subs( s[2], x );
```

$$X := \sqrt{3}$$

De vergelijking $e^x = 0$ heeft geen oplossing

```
> s := solve( exp(x)=0, x );
```

$$s :=$$

```
> a+s;
```

$$a+$$

Toelichting

De allereenvoudigste vorm waarin een `solve`-opdracht kan worden gegeven is `solve(p)`. Omdat `p` hier geen *vergelijking* is (er staat geen $=$-teken in), wordt dit automatisch geïnterpreteerd als `p=0`. En omdat er maar één variabele in de expressie `p` voorkomt, wordt de vergelijking automatisch voor deze variabele opgelost. De schrijfwijze `solve(p)` is dus in dit geval een afkorting voor `solve( p=0, x )`.
Let op het verschil dat we krijgen in de *vorm* van het antwoord als we de op te lossen variabele al of niet tussen accolades zetten.
De werkwijze `X := subs( s[2], x )` om een oplossing toe te kennen aan de variabele X lijkt nogal omslachtig: “substitueer $x = \sqrt{3}$ in de expressie x en ken het resultaat toe aan variabele X.” Het heeft verschillende voordelen, bij ingewikkelde antwoorden, maar vooral als er méér vergelijkingen zijn met verschillende oplossingen.
Voor geen enkele waarde van x is $e^x = 0$. Als we vragen om de oplossing(en) van deze vergelijking aan de variabele `s` toe te kennen, dan wordt `s` *helemaal niets*, dus niet eens meer een variabele zonder waarde, want in dat geval zou de opdracht `a+s;` moeten resulteren in $a + s$. ◇

Vaak zal een vergelijking, naast de ‘op te lossen variabele’ nog ‘onbekende constanten’ bevatten. Een intelligent mens is zich daar niet altijd van bewust, en dat kan hem misschien wel eens voor een verrassing plaatsen als hij zich door een computerprogramma laat assisteren.

Voorbeeldopgave

Kent Maple de '*abc*-formule' (wortelformule)?

Voorbeeldsessie

```
> p := a*x^2 + b*x + c;
```

$$p := ax^2 + bx + c$$

```
> solve( p=0 );
```

$$\{c = -ax^2 - bx, a = a, x = x, b = b\}$$

```
> solve( p=0, x );
```

$$-1/2\,\frac{b-\sqrt{b^2-4\,ac}}{a},\ -1/2\,\frac{b+\sqrt{b^2-4\,ac}}{a}$$

```
> solve( p=0, c );
```

$$-ax^2 - bx$$

Toelichting

Als we nalaten te vermelden dat x uit de vergelijking moet worden opgelost, beschouwt Maple het als een vergelijking in de onbekenden a, x, b en c. In dit geval heeft Maple ervoor gekozen om c uit te drukken in a, b en x. ◇

Als we bijvoorbeeld alleen de positieve oplossingen van een vergelijking of van een stelsel vergelijkingen willen hebben, dan is `assume` daar niet voor geschikt.

Voorbeeldsessie

```
> p := x^2+x-2;
```

$$p := x^2 + x - 2$$

```
> solve( {p = 0}, {x} ) assuming x>0;
```

$$\{x = 1\}, \{x = -2\}$$

```
> solve( {x^2+x-2, x>0}, {x} );
```

$$\{x = 1\}$$

Toelichting

Het `solve`-commando trekt zich dus blijkbaar niets aan van de aanname over de variabele `x`. Een uitweg is in dit geval om Maple een *stelsel* te laten oplossen met als eerste element de vergelijking en als tweede een *ongelijkheid*. ◇

Identieke vergelijkingen Een speciaal geval hebben we als we in een vergelijking de parameters willen bepalen, zó dat de vergelijking klopt voor alle waarden van x. Daarvoor kunnen we de functie `identity` gebruiken. Hoe dat gaat, laten we zien met een ietwat triviaal voorbeeld.

`identity`

Voorbeeldopgave

Voor welke waarden van a, b, c is $ax + b = c$ voor alle x?

Voorbeeldsessie

```
> vergelijking := a*x + b = c;
```

$$\mathit{vergelijking} := a\,x + b = c$$

```
> s1 := solve( vergelijking, {a,b} );
```

$$\mathit{s1} := \{a = -\frac{b-c}{x},\, b = b\}$$

```
> s2 := solve( identity(vergelijking, x), {a,b} );
```

$$\mathit{s2} := \{a = 0,\, b = c\}$$

Toelichting

Als a en b ‘gewoon’ uit de vergelijking wordt opgelost, vinden we een waarde voor a die afhankelijk is van x. Dat is dus niet de bedoeling. Maar als we vermelden dat de vergelijking moet worden opgevat als een identiteit in x, dan wordt de juiste oplossing gevonden. ◇

5.4 Oplossen van stelsels vergelijkingen

We kunnen in Maple *proberen* zo’n stelsel op te lossen met de procedure `solve`. Aan deze procedure moeten we normaliter twee dingen meegeven: een verzameling *vergelijkingen* en een verzameling *onbekenden*. Wanneer we de vergelijkingen `v1=0, ..., vn=0` willen oplossen naar de variabelen `x1, ..., xm` ($m \leq n$), dan gaat dit met

```
solve( {v1=0,...,vn=0}, {x1,...,xm} ).
```

Bij één vergelijking – waaruit dus ook maar één onbekende kan worden opgelost – mogen de accolades ook worden weggelaten.
Als het rechterlid van de vergelijkingen nul is, mogen we dit ook weglaten:

```
solve( {v1,...,vn}, {x1,...,xm} ).
```

Voor een stelsel van *lineaire* vergelijkingen kan beter gebruik gemaakt worden van de faciliteiten in de bibliotheek `LinearAlgebra`, zie Module 14.
Als `solve` geen enkele oplossing heeft gevonden, dan geeft Maple helemaal geen uitvoer, dus ook geen vermelding dat er geen oplossing is gevonden. Als er wel oplossingen zijn gevonden, komen deze in de vorm van een *rij* van *verzamelingen*. Het is verstandig het resultaat van `solve( )` toe te kennen aan een variabele, bijvoorbeeld aan de variabele `oplossing` door het commando

```
oplossing := solve( {v1=0,...,vn=0}, {x1,...,xm} );
```

Dan kunnen we de eerste oplossing krijgen met `oplossing[1]` enzovoort. Zie ook de bespreking van rijen (*sequences*) in Module 8. Elk van de oplossingen heeft de vorm

```
{ x1 = opl1, ..., xm = oplm }.
```

Nu zouden we de oplossing aan de variabelen kunnen toekennen met `x1 := opl1;, ..., xm := oplm;` maar dit is nogal omslachtig, omdat we elke keer de expressies `opl1` etc. moeten overtikken. Voor dit geval is `assign(oplossing[1]);` veel handiger (of, als er maar één oplossing is: `assign(oplossing);`).

`assign`

Maar in plaats van de oplossing toe te kennen aan de afzonderlijke variabelen, kunnen we ook de oplossing substitueren in een andere expressie waarin de namen van de betreffende variabelen voorkomen, door middel van

`subs`

```
subs( oplossing[1], expressie )
```

De oplossingen die `solve` geeft, staan al in de goede vorm voor substitutie!

Voorbeeldopgave

Los op het stelsel $x^2 + y = 3$, $x + y = 3$, en controleer elk van de oplossingen door substitutie in deze vergelijkingen.

Voorbeeldsessie

```
> f1 := x^2 + y = 3:     f2 := x + y = 3:
> opl := solve( {f1,f2}, {x,y} );
```

$$opl := \{y = 2, x = 1\}, \{y = 3, x = 0\}$$

```
> subs( opl[1], [f1,f2] );
```

$$[3 = 3, 3 = 3]$$

```
> subs( opl[2], [f1,f2] );
```

$$[3 = 3, 3 = 3]$$

De oplossingen staan in willekeurige volgorde. We krijgen eerst de x- en daarna de y-waarde van de tweede set oplossingen door:

```
> subs( opl[2], [x,y] );
```

$$[0, 3]$$

We kunnen dat zelfs in een lijstje 'vergelijkingen' zetten:

```
> subs( opl[2], [X=x,Y=y] );
```

$$[X = 0, Y = 3]$$

of er een zogenaamde 'tabel' van maken

```
> Opl := table(opl[1]);
```

$$Opl := table\,([y = 2, x = 1])$$

```
> Opl[x], Opl[y];
```

$$1, 2$$

Directe toekenning aan de variabelen x en y:

```
> assign(opl[1]);
```

met als resultaat

```
> x,y;
```

$$1, 2$$

zodat de oorspronkelijke vergelijkingen nu worden:

```
> [f1,f2];
```

$$[3 = 3, 3 = 3]$$

```
> assign(opl[2]);
```

Error, (in assign) invalid arguments

immers **opl[2]** ziet er nu zo uit:

```
> opl[2];
```

$$\{2 = 3, 1 = 0\}$$

Toelichting

Merk op dat aan de variabelen `f1` en `f2` *vergelijkingen* als 'waarde' wordt toegekend. Inderdaad voldoen de gevonden oplossingen aan

de vergelijkingen. Merk op dat voor het controleren hiervan `subs` handig werkt.
De twee oplossingen worden gegeven als *verzamelingen* (te herkennen aan de accolades) van x- en y-waarde; de *volgorde* is écht willekeurig. Als ze na een `restart` opnieuw worden berekend, is er een gerede kans dat ze er in een andere volgorde uit komen. Als ze, zoals in dit voorbeeld is gebeurd, tussen vierkante haakjes worden gezet, dan is de volgorde wél vast.
Een andere mogelijkheid is om er een *tabel* met de naam `Opl` van te maken. De x- en y-waarde van de gekozen oplossing zijn dan beschikbaar als `Opl[x]` en `Opl[y]`. Zie verder §8.7.
Het eenvoudigst is natuurlijk om een `assign`-opdracht te gebruiken. Een belangrijk nadeel hiervan is dat daardoor `x` en `y` geen 'vrije' variabelen meer zijn en een eventuele andere oplossing niet meer beschikbaar is. ◇

> **!** Gebruik `assign` om oplossingen 'in te vullen' alléén als er maar één oplossing is, én u zeker weet dat u de gehele verdere sessie met deze waarden verder wilt werken.

Als afsluiting van deze paragraaf volgen nog twee voorbeelden waarin het gebruik van `solve` en `subs` wordt geïllustreerd.

Voorbeeldopgave

Gevraagd een functie $p(x)$ van de vorm $ax^2 + bx + c$, waarvoor geldt: $p(1) = 1$, $p(2) = 6$ en $p(3) = 2$.

Voorbeeldsessie

We maken de functie p met onbekende a, b en c:

```
> p := x -> a*x^2 + b*x + c;
```

$$p := x \mapsto ax^2 + bx + c$$

We moeten a, b en c oplossen uit het volgende stelsel vergelijkingen:

```
> stelsel := {p(1)=1, p(2)=6, p(3)=2};
```

$$stelsel := \{a + b + c = 1, 4\,a + 2\,b + c = 6, 9\,a + 3\,b + c = 2\}$$

```
> s := solve( stelsel, {a,b,c} );
```

$$s := \left\{c = -13, b = \frac{37}{2}, a = -9/2\right\}$$

Substitueren in $p\,(x)$

```
> subs( s, p(x) );
```

$$-\frac{9}{2}\,x^2 + \frac{37}{2}\,x - 13$$

en er een functie van maken:

```
> pol := unapply( %, x );
```

$$pol := x \mapsto -\frac{9}{2}\,x^2 + \frac{37}{2}\,x - 13$$

Controle:

```
> pol(1), pol(2), pol(3);
```

$$1,\ 6,\ 2$$

Voorbeeldopgave

Elimineer λ en μ uit het volgende stelsel vergelijkingen:

$$\begin{cases} x = 1 - 2\lambda + 3\mu \\ y = 4 + \lambda - 5\mu \\ z = -3 + 4\lambda + 2\mu \end{cases}$$

(Dit is de parametervoorstelling van een vlak. Gevraagd wordt dus om een vergelijking voor dit vlak op te stellen.)

Voorbeeldsessie

```
> v1 := x = 1 -2*lambda + 3*mu:
  v2 := y = 4 + lambda - 5*mu:
  v3 := z = -3 + 4*lambda + 2*mu:
```

los λ en μ op uit de eerste twee vergelijkingen

```
> s := solve( {v1,v2}, {lambda,mu} );
```

$$s := \{\mu = -2/7\,y + \tfrac{9}{7} - 1/7\,x, \lambda = -5/7\,x + \tfrac{17}{7} - 3/7\,y\}$$

en substitueer het resultaat in de derde vergelijking

```
> subs( s, v3 );
```

$$z = \frac{65}{7} - \frac{22}{7}\,x - \frac{16}{7}\,y$$

als je geen breuken in de vergelijking wilt:

```
> vergelijking := 7*%;
```

$$\mathit{vergelijking} := 7\,z = 65 - 22\,x - 16\,y$$

Alternatief: Los λ, μ en z op

```
> s := solve( {v1,v2,v3}, {lambda,mu,z} );
```

$$s := \{\mu = -2/7\,y + \tfrac{9}{7} - 1/7\,x, \lambda = -5/7\,x + \tfrac{17}{7} - 3/7\,y, z = \tfrac{65}{7} - \tfrac{22}{7}\,x - \tfrac{16}{7}\,y\}$$

We weten nu niet zeker dat het derde element de gevraagde vergelijking is. Een truc om deze eruit te halen (zonder te zeggen dat het de derde is):

```
> vergelijking := z = subs(s,z);
```

$$\mathit{vergelijking} := z = \frac{65}{7} - \frac{22}{7}\,x - \frac{16}{7}\,y$$

5.5 RootOf

RootOf

Soms staat in het resultaat van `solve` een mededeling als

```
x = RootOf(_Z^2 + 1)
```

Dit betekent dat `x` een oplossing is van de vergelijking $\_Z^2 + 1 = 0$. (En niet dat `x` gelijk is aan $\sqrt{\_Z^2 + 1}$!) Ondanks dat `solve` deze oplossingen niet geeft, kunnen we hier vaak toch een en ander uit afleiden. Bijvoorbeeld in dit geval kunnen we concluderen dat x niet reëel is.
Stel dat `oplos` een resultaat van `solve` is waarin de uitdrukking

```
x = RootOf(_Z^2+1)
```

voorkomt. Dan representeert `oplos` in feite meer dan één oplossing, omdat de vergelijking $\_Z^2 + 1 = 0$ meer dan één oplossing heeft. Als Maple de nulpunten kan vinden van de expressie tussen de haken van de `RootOf`-procedure, dan kan de variabele `oplos` worden omgezet in een rij expressies met behulp van

allvalues

```
allvalues(oplos);
```

Elke expressie in deze rij wordt verkregen uit `oplos` door `x` te vervangen door een van de oplossingen van de vergelijking tussen de haken van de `RootOf`-procedure.

Voorbeeldopgave

Vind alle reële oplossingen van het stelsel

$$x^2 - y = 0, \qquad y^2 - x = 0\,.$$

Voorbeeldsessie

```
> vgln := {x^2-y=0, y^2-x=0}:
> solve( vgln, {x,y} );
```

$$\{y = 0,\, x = 0\},\, \{y = 1,\, x = 1\}, \{y = \mathrm{RootOf}(\_Z^2 + \_Z + 1),\; x = -1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)\}$$

```
> opl3 := %[3];
```

$$\mathit{opl3} := \{y = \mathrm{RootOf}(\_Z^2 + \_Z + 1),\; x = -1 - \mathrm{RootOf}(\_Z^2 + \_Z + 1)\}$$

```
> ad := allvalues(opl3);
```

$$ad := \{y = -\frac{1}{2} + \frac{1}{2}\, I\, \sqrt{3},\, x = -\frac{1}{2} - \frac{1}{2}\, I\, \sqrt{3}\},\; \{y = -\frac{1}{2} - \frac{1}{2}\, I\, \sqrt{3},\, x = -\frac{1}{2} + \frac{1}{2}\, I\, \sqrt{3}\}$$

Als men alleen reële oplossingen wil hebben:

```
> use RealDomain in solve(vgln,{x,y}) end use;
```

$$\{x = 0, y = 0\}, \{x = 1, y = 1\}$$

Toelichting

De eerste twee oplossingen ((0, 0) en (1, 1)) zijn niet verder onderzocht; de overige oplossingen komen in de `RootOf`-vorm. ◇

Als men alleen in reële oplossingen geïnteresseerd is, ligt het gebruik van `with(RealDomain)` voor de hand. Dit moet echter met enige argwaan gebeuren omdat Maple in `RealDomain` soms oplossingen ‘over het hoofd ziet’. In het bovenstaande voorbeeld gaat het goed met Maple 14. Vooral bij stelsels vergelijkingen is het verstandiger om Maple gewoon complex te laten rekenen en er zelf de reële oplossingen uit te pikken.

> **!** Gooi niet direct alle oplossingen met een `I` erin weg als u alleen reële oplossingen wilt hebben. Doe eerst even `evalc` bij zo’n oplossing; soms zal bijken dat zij na vereenvoudiging tóch reëel is.

Hoe complexe oplossingen ‘automatisch’ kunnen worden verwijderd wordt besproken in Module 8.

5.6 Numerieke benadering van oplossingen

`fsolve`

Het commando `fsolve` kan worden gebruikt voor het vinden van een numerieke benadering van een oplossing van n vergelijkingen met n onbekenden (dus precies even veel vergelijkingen als onbekenden). We bekijken eerst het geval van één vergelijking met één onbekende. Ook als de vergelijking meer dan één oplossing heeft, zal Maple er vaak maar één geven – en als men alleen `fsolve(vergelijking)` geeft, is van te voren niet te zeggen welke dat zal zijn. Als men bijvoorbeeld de oplossingen van de vergelijking $x^3 + 3x^2 - e^x = 1$ zou willen bepalen geeft Maple na

```
fsolve(x^3 + 3*x^2 - exp(x) = 1);
```

slechts één van de vier reële oplossingen.
Men kan aan `fsolve` als tweede argument opgeven tot welk interval Maple zich moet beperken bij het zoeken naar een oplossing. In dit voorbeeld zou men een positieve oplossing kunnen benaderen met

```
fsolve(x^3 + 3*x^2 - exp(x) = 1, x=0..1);
```

Hierbij geeft `x=0..1` aan dat in het interval $[0,1]$ gezocht moet worden. Als men *alle* (reële!) oplossingen zou willen vinden, kan men het best een grafiek tekenen van $x^3+3x^2-e^x-1$ (zie Module 9), nagaan waar de nulpunten ongeveer liggen en steeds `fsolve` opgeven met een interval waarvan men zeker weet dat het maar één oplossing bevat.

`avoid` Een andere mogelijkheid is het gebruik van de optie `avoid` om aan te geven dat Maple triviale (bijvoorbeeld $x=0$) of eerder gevonden oplossingen niet hoeft te berekenen.

Voorbeeldopgave

Bepaal numerieke benaderingen van de oplossingen van de volgende vergelijkingen:
(1) $x^3+3x^2-x=1$; (2) $x^3+3x^2-e^x=1$.

Voorbeeldsessie

```
> vergelijking := x^3 + 3*x^2 - x = 1;
```

$$vergelijking := x^3 + 3\,x^2 - x = 1$$

```
> fsolve(vergelijking);
```

$$-3.214319743,\ -0.4608111272,\ 0.6751308706$$

```
> f := x -> x^3 + 3*x^2 - exp(x);
```

$$f := x \mapsto x^3 + 3\,x^2 - e^x$$

```
> x1 := fsolve( f(x)=1, {x} );
```

$$x1 := \{x = 0.9535687636\}$$

Er zijn meer oplossingen, bijvoorbeeld

```
> fsolve( f(x) = 1, {x}, 5..6 );
```

$$\{x = 5.587757129\}$$

```
> x2 := fsolve( f(x)=1, {x}, avoid={x1} );
```

$$x2 := \{x = -0.8123968060\}$$

```
> x3 := fsolve( f(x)=1, {x}, avoid={x1,x2} );
```

$$x3 := \{x = -2.871893711\}$$

```
> x4 := fsolve( f(x)=1, {x}, avoid={x1,x2,x3} );
```

$$x4 := \{x = 5.587757129\}$$

```
> x5 := fsolve( f(x)=1, {x}, avoid={x1,x2,x3,x4} );
```

$$x5 := \mathit{fsolve}(x^3 + 3\,x^2 - e^x = 1, \{x\}, \mathit{avoid} = \{\{x = -0.8123968060\},$$
$$\{x = 0.9535687636\}, \{x = -2.871893711, \{x = 5.587757129\}\})$$

Toelichting

De eerste vergelijking is een *polynoom* en Maple vindt direct alle oplossingen.
De tweede vergelijking is problematischer wegens de term e^x. Maple geeft dan ook maar één oplossing. Dat er méér zijn blijkt als we Maple vragen om zich te beperken tot het interval $[5, 6]$ (in feite hebben we eerst een grafiek van $f(x) - 1$ getekend en geconstateerd dat deze tussen $x = 5$ en $x = 6$ nog een nulpunt heeft). Met de `avoid`-optie lukt het Maple om vier oplossingen te vinden, maar dan houdt het op. Om er zeker van te zijn dat er niet meer oplossingen zijn moeten andere middelen worden ingezet. ◇

Voor het numeriek oplossen van een stelsel van meer vergelijkingen met meer onbekenden kan `fsolve` ook worden gebruikt:

```
fsolve( {stelsel}, {onbekenden} );
```

vindt echter altijd maar één reële oplossing. Om Maple te instrueren dat het in de buurt van een gegeven punt moet zoeken kunt u dat punt opgeven door

```
fsolve( {f(x,y)=0,g(x,y)=0}, {x=1,y=2} );
```

voor een stelsel van twee vergelijkingen met twee onbekenden. Op een analoge manier is het mogelijk het gebied af te bakenen door intervallen voor de onbekenden op te geven in de vorm `x=0..2`, enzovoort.

Opgave 5.1

Bepaal alle oplossingen van het stelsel $yz = 1$, $yx = 1$, $xz = 1$.

Opgave 5.2

Bepaal alle oplossingen van het stelsel

$$\left\{ \begin{array}{rcl} -9x^2 + 10x + 12xy - 8y & = & 0 \\ 3x^2 - 4x - 4y & = & 0 \end{array} \right.$$

Controleer de antwoorden door ze in te vullen in de vergelijkingen.

Opgave 5.3

Bepaal de *exacte* oplossingen van de vergelijking $x^3 + 3x^2 - x = 1$. Zijn ze alle drie reëel? (zie de voorbeeldsessie in §5.6)

Opgave 5.4

Gegeven het stelsel van drie vergelijkingen in de onbekenden x, y, z, u, v:

$$\begin{cases} x = u(1-v) \\ y = uv \\ z = (1-u)v \end{cases}$$

Vind een uitdrukking voor z als functie van x en y door u en v uit het stelsel te elimineren.

Opgave 5.5

Bepaal een functie f van de vorm $a + b\,e^{cx}$ waarvoor geldt

(a) $f(0) = 1, \quad f(1) = 2, \quad f(2) = 5$;
(b) $f(0) = 1, \quad f(1) = 2, \quad f(3) = 5$.

Aanwijzing bij (b): Er is maar één oplossing reëel.

Opgave 5.6

Bepaal alle oplossingen van de vergelijking $\cos(x) = 0$. Geeft `solve` alle oplossingen? Zou `solve` zomaar gebruikt kunnen worden om de inverse te bepalen van de transformatie naar poolcoördinaten:

$$x = r\cos(p), \quad y = r\sin(p)\,?$$

Opgave 5.7

Los z op uit de vergelijking

$$e^z = -4\sqrt{3} + 4\,i$$

Vindt Maple alle oplossingen?

Opgave 5.8

Bereken alle oplossingen van de volgende vergelijkingen:

(a) $z^4 = -16\,i$
(b) $z^4 + 2z^2 + 2 = 0$

Geef de oplossingen zowel in Cartesische als in polaire vorm.
Aanwijzing: Soms werkt `convert(z,exp)` beter dan het commando `convert(z,polar)`. Gebruik zo nodig `radnormal`.